



1st EuroProofNet Dedukti School

Introduction
to proof system interoperability,
the Dedukti language
and the Lambdapi tool

Frédéric Blanqui

DeducTeam

Inria

école
normale
supérieure
paris-saclay



24 June 2022

Thank you

Nicolas Tabareau, Matthieu Sozeau and their colleagues
for the local organization in Nantes of
Women in EuroProofnet and the 1st Dedukti school !

Outline

Introduction to proof system interoperability

$\lambda\Pi$ -calculus modulo rewriting


Dedukti language

Lambdapi tool

Libraries of formal proofs today

Library	Nb files	Nb objects*
Coq Opam	16,000	473,000
Isabelle AFP	7,000	90,000
Lean Mathlib	2,000	81,000
Mizar Mathlib	1,400	77,000
HOL-Light	500	35,000
...


* type, definition, theorem, ...



Libraries of formal proofs today

Library	Nb files	Nb objects*
Coq Opam	16,000	473,000
Isabelle AFP	7,000	90,000
Lean Mathlib	2,000	81,000
Mizar Mathlib	1,400	77,000
HOL-Light	500	35,000
...

* type, definition, theorem, ...




- ▶ Every system has basic libraries on integers, lists, ...
- ▶ Some definitions/theorems are available in one system only

Libraries of formal proofs today

Library	Nb files	Nb objects*
Coq Opam	16,000	473,000
Isabelle AFP	7,000	90,000
Lean Mathlib	2,000	81,000
Mizar Mathlib	1,400	77,000
HOL-Light	500	35,000
...

* type, definition, theorem, ...




- ▶ Every system has basic libraries on integers, lists, ...
 - ▶ Some definitions/theorems are available in one system only
- ⇒ Can't we translate a proof between two systems automatically?

Interest of proof interoperability

- ▶ Avoid duplicating developments and losing time
- ▶ Facilitate development of new proof systems
- ▶ Increase reliability of formal proofs (cross-checking)
- ▶ Facilitate validation by certification authorities
- ▶ Relativize the choice of a system (school, industry)
- ▶ Provide multi-system data to machine learning


Difficulties of interoperability

- ▶ Each system is based on different axioms and deduction rules
- ▶ It is usually non trivial and sometimes impossible to translate a proof from one system to the other (e.g. a classical proof in an intuitionistic system)
- ▶ Is it reasonable to have $n(n - 1)$ translators for n systems?



Difficulties of interoperability

- ▶ Each system is based on different axioms and deduction rules
- ▶ It is usually non trivial and sometimes impossible to translate a proof from one system to the other (e.g. a classical proof in an intuitionistic system)
- ▶ Is it reasonable to have $n(n - 1)$ translators for n systems?



A common language for proof systems?

Logical framework D

language for describing axioms, deduction rules and proofs of a system S as a theory $D(S)$ in D

Example: $D =$ predicate calculus

allows one to represent $S =$ geometry, $S =$ arithmetic, $S =$ set theory, ...
not well suited for functional computations and dependent types

A common language for proof systems?

Logical framework D

language for describing axioms, deduction rules and proofs of a system S as a theory $D(S)$ in D

Example: $D =$ predicate calculus

allows one to represent $S =$ geometry, $S =$ arithmetic, $S =$ set theory, ...
not well suited for functional computations and dependent types

Better: $D = \lambda\Pi$ -calculus modulo rewriting


allows one to represent also:

$S =$ HOL, $S =$ Coq, $S =$ Agda, $S =$ PVS, ...

How to translate a proof $t \in A$ in a proof $u \in B$?


In a logical framework D :

system A



$D(A)$

$D(B)$




system B

1. translate $t \in A$ in $t' \in D(A)$

3. translate $u' \in D(B)$ in $u \in B$

How to translate a proof $t \in A$ in a proof $u \in B$?


In a logical framework D :



1. translate $t \in A$ in $t' \in D(A)$
2. identify the axioms and deduction rules of A used in t'
translate $t' \in D(A)$ in $u' \in D(B)$ if possible
3. translate $u' \in D(B)$ in $u \in B$

How to translate a proof $t \in A$ in a proof $u \in B$?

In a logical framework D :



1. translate $t \in A$ in $t' \in D(A)$
2. identify the axioms and deduction rules of A used in t'
translate $t' \in D(A)$ in $u' \in D(B)$ if possible
3. translate $u' \in D(B)$ in $u \in B$

\Rightarrow represent in the same way functionalities common to A and B

The modular $\lambda\Pi/\mathcal{R}$ theory U and its sub-theories

38 symbols, 28 rules, 13 sub-theories

0
succ
pred
positive

$Prf_c, \Rightarrow_c, \wedge_c, \vee_c, \forall_c, \exists_c$

$\top, \perp, \neg, \wedge, \vee, \exists$

\Rightarrow

\forall

$Set, El, \iota, Prop, Prf$

\sim_d

\Rightarrow_d


π

o


\sim

$Scheme, Els, \uparrow, \forall, \forall$


$psub, pair, fst, snd, pair^\dagger$



Dedukti, an assembly language for proof systems



Libraries currently available in Dedukti


System	Libraries
HOL-Light	OpenTheory
Matita	Arith
Coq	Stdlib parts, GeoCoq
Isabelle	HOL.Complex_Main  (AFP soon?)
Agda	Stdlib parts ($\pm 25\%$)
PVS	Stdlib parts
TPTP	E 69%, Vampire 83%

Case study:


Matita/Arith \longrightarrow OpenTheory, Coq, PVS, Lean, Agda

<http://logipedia.inria.fr>

Functionalities



Today



Outline

Introduction to proof system interoperability

$\lambda\Pi$ -calculus modulo rewriting

Dedukti language

Lambdapi tool

What is the $\lambda\Pi$ -calculus modulo rewriting?

$\lambda\Pi/\mathcal{R} =$

λ

simply-typed λ -calculus

+ Π

dependent types, e.g. $\text{array}(n)$

+ \mathcal{R}

identification of types modulo rewrites rules $l \hookrightarrow r$

What is the $\lambda\Pi$ -calculus modulo rewriting?

$\lambda\Pi/\mathcal{R} =$

λ simply-typed λ -calculus
 $+ \Pi$ dependent types, e.g. $\text{array}(n)$
 $+ \mathcal{R}$ identification of types modulo rewrites $l \hookrightarrow r$

terms $t, u =$

TYPE sort of types
 f global constant
 x local variable
 tu application
 $\lambda x : t, u$ abstraction
 $\Pi x : t, u$ dependent product
 $t \rightarrow u$ abbreviation for $\Pi x : t, u$ when $x \notin u$

What is the $\lambda\Pi$ -calculus modulo rewriting?

theory =

Σ sequence of type declarations for global constants
+ \mathcal{R} set of rewrite rules $l \hookrightarrow r$
 including rules on types !

What is the $\lambda\Pi$ -calculus modulo rewriting?

theory =

Σ sequence of type declarations for global constants
+ \mathcal{R} set of rewrite rules $l \hookrightarrow r$
 including rules on types !

typing = ... +

$$\frac{\Gamma, x : A \vdash t : B \quad \Gamma \vdash \Pi x : A, B : \text{TYPE}}{\Gamma \vdash \lambda x : A, t : \Pi x : A, B}$$

Γ : types of
local variables

$$\frac{\Gamma \vdash t : \Pi x : A, B \quad \Gamma \vdash u : A}{\Gamma \vdash tu : B\{x \mapsto u\}}$$

$$\frac{\Gamma \vdash t : A \quad A \equiv_{\beta\mathcal{R}} B}{\Gamma \vdash t : B}$$

$\equiv_{\beta\mathcal{R}}$: equational theory
generated by β and \mathcal{R}

ex. concat : $\Pi p : \mathbb{N}, \text{array } p \rightarrow \Pi q : \mathbb{N}, \text{array } q \rightarrow \text{array}(p + q)$
concat 2 : $\text{array } 2 \rightarrow \Pi q : \mathbb{N}, \text{array } q \rightarrow \text{array}(2 + q)$

Properties of the $\lambda\Pi$ -calculus modulo rewriting

$\lambda\Pi/\mathcal{R}$ enjoys all the properties of $\lambda\Pi$:

- ▶ unicity of types modulo $\equiv_{\beta\mathcal{R}}$
- ▶ decidability of $\equiv_{\beta\mathcal{R}}$ and type-checking

assuming that $\hookrightarrow_{\beta\mathcal{R}}$:

- ▶ terminates: there is no infinite $\hookrightarrow_{\beta\mathcal{R}}$ sequences
- ▶ is confluent: the order of $\hookrightarrow_{\beta\mathcal{R}}$ steps does not matter
- ▶ \mathcal{R} preserves typing: if $l\theta : A$ and $l \hookrightarrow r \in \mathcal{R}$ then $r\theta : A$

There exists (certified) tools for checking those properties

Outline

Introduction to proof system interoperability

$\lambda\Pi$ -calculus modulo rewriting

Dedukti language

Lambdapi tool

What is Dedukti?

Dedukti is a concrete language for defining $\lambda\Pi/\mathcal{R}$ theories

There are several tools to check the correctness of Dedukti files:

- ▶ Kocheck <https://github.com/01mf02/kontroli-rs>
- ▶ Dkcheck <https://github.com/Deducteam/dedukti>
- ▶ Lambdapi <https://github.com/Deducteam/lambdapi>

Efficiency: Kocheck > Dkcheck > Lambdapi

Features: Kocheck < Dkcheck < Lambdapi

Dkcheck and Lambdapi can export $\lambda\Pi/\mathcal{R}$ theories to:

- ▶ the HRS format of the confluence competition
- ▶ the XTC format of the termination competition extended with dependent types

How to install and use Kocheck?

Installation:

```
cargo install --git https://github.com/01mf02/kontroli-rs
```

Use:

```
kocheck file.dk
```

How to install and use Dkcheck?

Installation:

Using Opam:

```
opam install dedukti
```

Compilation from the sources:

```
git clone https://github.com/Deducteam/dedukti.git  
cd dedukti  
make  
make install
```

Use:

```
dk check file.dk
```

Dedukti syntax

BNF grammar:

<https://github.com/Deducteam/Dedukti/blob/master/syntax.bnf>

file extension: .dk

comments: (; ... (; ... ;) ... ;)

identifiers:

(a-z|A-Z|0-9|_)+ and { | arbitrary string | }

Terms

Type

id

id.id

term term ... term

id [: term] => term

[id :] term -> term

(term)

sort for types

variable or constant

constant from another file

application

abstraction

[dependent] product

Command for declaring/defining a symbol

*modifier** *id* *param** : *term* [:= *term*] .

param ::= (*id* : *term*)

modifier's:

- ▶ **def**: definable
- ▶ **thm**: never reduced
- ▶ **AC**: associative and commutative
- ▶ **private**: exported but usable in rule left-hand sides only
- ▶ **injective**: used in subject reduction

```
N : Type .
0 : N .
s : N -> N .
def add : N -> N -> N .

thm add_com :
  x:N -> y:N -> Eq (add x y) (add y x) := ...
```


Command for declaring rewrite rules

$[id *] (term \rightarrow term)^+ .$

```
[x y]
x + 0 --> x
x + s y --> s (x + y).
```

Dkcheck tries to automatically check:

preservation of typing by rewrite rules (aka subject reduction)

Queries and assertions

INFER *term* .

EVAL *term* .

(# ASSERT | # ASSERTNOT) *term* (:|==) *term* .

(# CHECK | # CHECKNOT) *term* (:|==) *term* .

```
# INFER 0.
```

```
# EVAL add 2 2.
```

```
# ASSERT 0 : N.
```

```
# ASSERTNOT 0 : N → N.
```

```
# ASSERT add 2 2 == 4.
```

```
# ASSERTNOT add 2 2 == 5.
```

Importing the declarations of other files

file1.dk:

```
A : Type.
```

file2.dk:

```
#REQUIRE file1.  
a : file1.A.
```

Outline

Introduction to proof system interoperability

$\lambda\Pi$ -calculus modulo rewriting

Dedukti language

Lambdapi tool

What is Lambdapi?

Lambdapi is an *interactive proof assistant* for $\lambda\Pi/\mathcal{R}$

- ▶ has its own syntax and file extension `.lp`
- ▶ can read and output `.dk` files
- ▶ symbols can have implicit arguments
- ▶ symbol declaration/definition generates typing/unification goals
- ▶ goals can be solved by structured proof scripts (tactic trees)
- ▶ ...

Where to find Lambdapi?

Webpage: <https://github.com/Deducteam/lambdapi>

User manual: <https://lambdapi.readthedocs.io/>

Libraries:

<https://github.com/Deducteam/opam-lambdapi-repository>

How to install Lambdapi?

2 possibilities:

1. Using Opam:

```
opam install lambdapi
```

2. Compilation from the sources:

```
git clone https://github.com/Deducteam/lambdapi.git  
cd lambdapi  
make  
make install
```

How to use Lambdapi?

2 possibilities:

1. Command line (batch mode):

```
lambdapi check file.lp
```

2. Through an editor (interactive mode):

- ▶ Emacs
- ▶ VSCode

Lambdapi automatically (re)compiles dependencies if necessary

How to install the Emacs interface?

3 possibilities:

1. Nothing to do when installing Lambdapi with opam
2. From Emacs using MELPA:

```
M-x package-install RET lambdapi-mode
```

3. From sources:

```
make install_emacs
```

+ add in ~/.emacs:

```
(load "lambdapi-site-file")
```

Emacs interface

```
emacs@bianqui-Latitude-5500
File Edit Options Buffers Tools Flymake Help
-- [Home]
/* We are now ready to prove that, for any natural number "x",
"zero + x" is equal to "x", that is, to show that there exists a term, that
we will call "zero_is_neutral_for_+", of type "0 x : N, Prf (zero + x = x)".
To this end, LambdaPI provides an interactive mode (launched by the keyword
"begin") to enable users to define this term step by step using tactics. */
opaque /* We declare the symbol as opaque as we do not want LambdaPI
to unfold it later. */
symbol zero_is_neutral_for_+ : Prf (zero + x = x) =
begin
/* Here, in Emacs or VSCode, the system prints the following goal to prove:
"zero_is_neutral_for_+ 0 x : N, Prf (zero + x = x) */
--
/* To proceed by induction on x, we simply need to say that
"zero_is_neutral_for_+" should be of the form "ind N _ _ _"
where "-" stands for a term to be built.
This can be done by using the "refine" tactic: */
/* However, if we simply write "refine ind N _ _ _",
LambdaPI will complain with the following error message:
"Missing subproofs (0 subproofs for 2 subgoals).".
This is because we gave no subproof for the case zero and case succ arguments.
Indeed, in LambdaPI, proofs must be well structured, that is, a tactic
must be followed by as many subproofs enclosed between curly brackets
as the number of subgoals generated by the tactic. So, here, we need to write
"refine nat _ _ _ () {}" and then write the missing subproofs. */
/* Remark: if we hadn't declared Prf as inductive, we would have gotten
4 subgoals: the first generated subgoal would not have been a typing goal
but the following unification goal:
LambdaPI[... tutorial.lp 4% (154.0) git-master (LambdaPI Flymake[0 20] 0
-- [Log]
74: 0 x : N, Prf (0 + x = x)
-- [Goals]
0/*- "Goals" All (1,0) (Fundamental company)
-- [Messages]
[file:///home/bianqui/src/lambdapi/tests/OK/tutorial.lp:37:0-20]
Cannot solve % = % = %.
0
[file:///home/bianqui/src/lambdapi/tests/OK/tutorial.lp:69:12-14]
Pattern variable p can be replaced by a
Loading "/home/bianqui/src/lambdapi/tests/OK/logic.lp" ...
Loading "/home/bianqui/src/lambdapi/tests/OK/bool.lp" ...
Loading "/home/bianqui/src/lambdapi/tests/OK/nat.lp" ...
[file:///home/bianqui/src/lambdapi/tests/OK/nat.lp:52:0-35:30]
Impossible critical pair:
t = s 00 + s s1
t [] 0 00 + (s 00 + s1') =* s (00 + (s1' + (00 + s1')))
with 00' + s s1' = 00' + (00 + s1')
s [] 0 s1 + (00 + s s1') =* s (s1' + (00 + (00 + (00 + s1'))))
with s 00 + s1 = s1 + (00 + s1)
```

checked part

edition buffer

goals

messages

window layout
can be customized

shortcuts: <https://lambdapi.readthedocs.io/en/latest/emacs.html>

How to install the VSCode interface?

From the VSCode Marketplace

VSCode interface

The image shows the Visual Studio Code interface with a Lean code editor, a goals panel, and a terminal window.

checked part

```
301 builtin "T" = T;
302 builtin "eq" = =;
303 builtin "refl" = ==refl;
304 builtin "eqind" = ==ind;
305
306 /* We now reproduce our theorem on the inductive type Nat instead of N,
307 using the tactics "induction", "reflexivity" and "rewrite".
308 To this end, we first need to define addition on Nat: */
309
310 symbol + : Nat → Nat → Nat;
311 notation + infix right 10;
312 rule $x + z ~ $x
313 with $x + s $y = s ($x + $y);
314
315 opaque symbol zero_is_neutral_for_+ x : Prf(z + x = x) =
316 begin
317   induction
318   { simplify; reflexivity; }
319   { assume x hyp_on_x; simplify; rewrite hyp_on_x; reflexivity; }
320 end;
321
```

goals

```
0 : Prf ((z + z) = z)
1 : ∀ x0 : Nat, Prf ((z + x0) = x0) → Prf ((z + s x0) = s x0)
```

edition buffer

messages

```
rule ind_Nat $0 $1 $2 z ~ $1
with ind_Nat $0 $1 $2 (s $3) ~ $2 $3 (ind_Nat $0 $1 $2 $3);
symbol ten : N
= 10;
symbol + : N → N → N;
notation + infix right associative 10.000000;
rule $0 + 0 ~ $0
with $0 + succ $1 ~ succ ($0 + $1)
with 0 + $0 ~ $0;
```

File `lambdapi.pkg`

developments must have a file `lambdapi.pkg` describing where to install the files relatively to the root of all installed libraries

```
package_name = my_lib  
root_path = logical.path.from.root.to.my_lib
```

Importing the declarations of other files

lambdapi.pkg:

```
package_name = unary  
root_path = nat.unary
```

file1.lp:

```
symbol A : TYPE;
```

file2.lp:

```
require nat.unary.file1;  
symbol a : nat.unary.file1.A;  
open nat.unary.file1;  
symbol a' : A;
```

file3.lp:

```
require open nat.unary.file1 nat.unary.file2;  
symbol b := a;
```

Lambdapi syntax

BNF grammar:

<https://raw.githubusercontent.com/Deducteam/lambdapi/master/doc/lambdapi.bnf>

file extension: `.lp`

comments: `/* ... /*... */... */` or `// ...`

identifiers: UTF16 characters and `{| arbitrary string |}`

Terms

TYPE

$(id.) * id$

$term \ term \ \dots \ term$

$\lambda \ id \ [: \ term] \ , \ term$

$\Pi \ id \ [: \ term] \ , \ term$

$term \ \rightarrow \ term$

-

$\text{let } id \ [: \ term] \ := \ term \ \text{in } term$

$(\ term \)$

sort for types

variable or constant

application

abstraction

dependent product

non-dependent product

unknown term

Command for declaring/defining a symbol

*modifier** *symbol* *id* *param** [: *term*] [:= *term*] [*begin proof end*] ;

param = *id* | _ | (*id*⁺ : *term*) | [*id*⁺ : *term*]
implicit
parameters

modifier's:

- ▶ *constant*: not definable
- ▶ *opaque*: never reduced
- ▶ *associative*
- ▶ *commutative*
- ▶ *private*: not exported
- ▶ *protected*: exported but usable in rule left-hand sides only
- ▶ *sequential*: reduction strategy
- ▶ *injective*: used in unification

Examples of symbol declarations

```
symbol N : TYPE;  
symbol 0 : N;  
symbol s : N → N;  
  
symbol + : N → N → N; notation + infix right 10;  
  
symbol × : N → N → N; notation × infix right 20;
```

Command for declaring rewrite rules

```
rule term  $\hookrightarrow$  term (with term  $\hookrightarrow$  term)* ;
```

pattern variables must be prefixed by \$:

```
rule $x + 0  $\hookrightarrow$  $x  
with $x + s $y  $\hookrightarrow$  s ($x + $y);
```

Lambdapi tries to automatically check:

preservation of typing by rewrite rules (aka subject reduction)

Command for adding rewrite rules

Lambdapi supports:

overlapping rules

```
rule $x + 0 ↦ $x  
with $x + s $y ↦ s ($x + $y)  
with 0 + $x ↦ $x  
with s $x + $y ↦ s ($x + $y);
```

matching on defined symbols

```
rule ($x + $y) + $z ↦ $x + ($y + $z);
```

non-linear patterns

```
rule $x - $x ↦ 0;
```

Lambdapi tries to automatically check:

local confluence (AC symbols/HO patterns not handled yet)

Higher-order pattern-matching

```
symbol R:TYPE;

symbol 0:R;
symbol sin:R → R;
symbol cos:R → R;
symbol D:(R → R) → (R → R);

rule D (λ x, sin $F.[x])
  ⇔ λ x, D $F.[x] × cos $F.[x];
rule D (λ x, $V.[])
  ⇔ λ x, 0;
```

Non-linear matching

Example: decision procedure for group theory

```
symbol G : TYPE;
symbol 1 : G;
symbol · : G → G → G; notation · infix 10;
symbol inv : G → G;

rule ($x · $y) · $z ↔ $x · ($y · $z)
with 1 · $x ↔ $x
with $x · 1 ↔ $x
with inv $x · $x ↔ 1
with $x · inv $x ↔ 1
with inv $x · ($x · $y) ↔ $y
with $x · (inv $x · $y) ↔ $y
with inv 1 ↔ 1
with inv (inv $x) ↔ $x
with inv ($x · $y) ↔ inv $y · inv $x;
```

Defining inductive-recursive types

because symbol and rule declarations are separated, one can easily define inductive-recursive types in Dedukti or Lambdapi:

```
// lists without duplicated elements  
  
constant symbol L : TYPE;  
  
symbol  $\notin$  :  $N \rightarrow L \rightarrow \text{Prop}$ ; notation  $\notin$  infix 20;  
  
constant symbol nil : L;  
constant symbol cons x l :  $\text{Prf}(x \notin l) \rightarrow L$ ;  
  
rule _  $\notin$  nil  $\leftrightarrow$  T  
with $x  $\notin$  cons $y $l _  $\leftrightarrow$  $x  $\neq$  $y  $\wedge$  $x  $\notin$  $l;
```

Command for generating induction principles

```
inductive N : TYPE := 0 : N | s : N → N;
```

is equivalent to:

```
symbol N : TYPE;  
symbol 0 : N;  
symbol s : N → N;  
symbol ind_N (p : N → Prop)  
  (case_0: Prf(p 0))  
  (case_s: Π x : N, Prf(p x) → Prf(p(s x)))  
  (n : N) : Prf(p n);  
rule ind_N $p $c0 $cs 0 ↔ $c0  
with ind_N $p $c0 $cs (s $x)  
  ↔ $cs $x (ind_N $p $c0 $cs $x)
```

Lambdapi handles strictly positive parametric inductive types

Example of inductive-inductive type

```
/* contexts and types in dependent type theory  
Forsberg's 2013 PhD thesis */
```

```
// contexts
```

```
inductive Ctx : TYPE :=
```

```
| □ : Ctx
```

```
| · Γ : Ty Γ → Ctx
```

```
// types
```

```
with Ty : Ctx → TYPE :=
```

```
| U Γ : Ty Γ
```

```
| P Γ a : Ty (· Γ a) → Ty Γ;
```

Queries and assertions

```
print id ;  
type term ;  
compute term ;  
(assert | assertnot) id * ⊢ term (:|≡) term ;
```

```
print N; // constructors and induction principle  
print +; // type and rules
```

```
type ×;  
compute 2 × 5;
```

```
assert 0 : N;  
assertnot 0 : N → N;
```

```
assert x y z ⊢ x + y × z ≡ x + (y × z);  
assertnot x y z ⊢ x + y × z ≡ (x + y) × z;
```

Reducing proof checking to type checking

(aka the Curry-Howard isomorphism)

```
// type of propositions
symbol Prop : TYPE;
symbol = : N → N → Prop; notation = infix 1;

// interpretation of propositions as types
// (Curry-Howard isomorphism)
symbol Prf : Prop → TYPE;

// examples of axioms
symbol --refl x : Prf(x = x);
symbol --s x y : Prf(x = y) → Prf(s x = s y);
symbol ind_N (p : N → Prop)
  (case_0: Prf(p 0))
  (case_s:  $\prod x : N, \text{Prf}(p x) \rightarrow \text{Prf}(p(s x))$ )
  (n : N) : Prf(p n);
```

Stating an axiom vs Proving a theorem

Stating an axiom:

```
opaque symbol 0_is_neutral_for_+ x :  
  Prf (0 + x = x);  
// no definition given now  
// one can still be given later with a rule
```

Proving a theorem:

```
opaque symbol 0_is_neutral_for_+ x :  
  Prf (0 + x = x) :=  
// generates the typing goal Prf (0 + x = x)  
// a proof must be given now  
begin  
  ... // proof script  
end;
```

Goals and proofs

symbol declarations/definitions can generate:

- ▶ typing goals $x_1 : A_1, \dots, x_n : A_n \vdash ? : B$
- ▶ unification goals $x_1 : A_1, \dots, x_n : A_n \vdash t \equiv u$

these goals can be solved by writing *proof*'s:

$$\begin{aligned} \textit{proof} &::= (\textit{proof_step} \ ;)^* \\ \textit{proof_step} &::= \textit{tactic} (\{ \textit{proof} \})^* \end{aligned}$$

- ▶ a *proof* is a ;-separated sequence of *proof_step*'s
- ▶ a *proof_step* is a *tactic* followed by as many *proof*'s enclosed in curly braces as the number of goals generated by the *tactic*

tactic's for unification goals:

- ▶ `solve` (applied automatically)

Example of proof

<https://raw.githubusercontent.com/Deducteam/lambdapi/master/tests/OK/tutorial.lp>

```
opaque symbol 0_is_neutral_for_+ x :  
  Prf(0 + x = x)  
:= begin  
  induction  
  {simplify; reflexivity;}  
  {assume x h; simplify; rewrite h; reflexivity;}  
end;
```

Tactics for typing goals

- ▶ `simplify` *[id]*
- ▶ `refine` *term*
 - ▶ `assume` *id*⁺
 - ▶ `generalize` *id*
 - ▶ `apply` *term*
 - ▶ `induction`
 - ▶ `have` *id* : *term*
 - ▶ `reflexivity`
 - ▶ `symmetry`
 - ▶ `rewrite` [*right*] [*pattern*] *term*
- ▶ `why3`

like Coq SSReflect
calls external provers

Lambdapi's additional features wrt Dkcheck/Kocheck

Lambdapi is an *interactive* proof assistant for $\lambda\Pi/\mathcal{R}$

- ▶ has its own syntax and file extension `lp`
- ▶ can read and output `dk` files
- ▶ supports Unicode characters and infix operators
- ▶ symbols can have implicit arguments
- ▶ symbol declaration/definition generates typing/unification goals
- ▶ goals can be solved by structured proof scripts (tactic trees)
- ▶ provides a `rewrite` tactic similar to Coq/SSReflect
- ▶ can call external (first-order) theorem provers
- ▶ provides a command for generating induction principles
- ▶ provides a local confluence checker
- ▶ handles associative-commutative symbols differently
- ▶ supports user-defined unification rules